媒介変数を用いたグラフィックス

Masato Shimura
JCD02773@nifty.ne.jp

2019 年 5 月 17 日

目次

1 円の公式

円の公式は幾つかあるが代表的なものは次の通り

1. 陰関数方程式

$$x^2 + y^2 = r^2$$

2. 極形式

$$z = r(cos\theta + isin\theta)$$

3. 媒介変数 (パラメーター) 表示

$$\begin{pmatrix} x \\ y \end{pmatrix} = \begin{cases} rcos\theta \\ rsin\theta \end{cases}$$

(1) は数式は奇麗でも格闘してもなかなか図にならない。数式マニアに委ねておこう

(2) はオイラーの美しい公式

$$e^{i\theta} = cos\theta + isin\theta$$

の左辺で計算する。

J には $e^{i\theta}$ を計算する r.(Angle) が備わっている。

```
 plot  r. 2p1 * (i.360)%360
```

360 角形を円としている



(3) は次のスクリプトで表現できる。ボックス内で x,y を各々計算しそのままグラフィックスに流し込む。

```
plot (cos ; sin) 2p1 * (i.360)%360
```

2　媒介変数

媒介変数=parameter

最後の parameter=媒介変数を用いると、複雑な曲線をシンプルに表現できる場合がある。有名な数学関数もあればグラフィックス向きの関数もある。

次に詳細なパラメータを用いたグラフィックスの一大コレクションがある。

3 態の式は Cartesian equation,Polar equation, parametrical equation と表記されているが図毎に 3 つの式がそろっているわけではない。

`http://www-history.mcs.st-and.ac.uk/Curves/Curves.html`

1. サイクロイド曲線

   サイクロイド曲線はヨハン・ベルヌイが提起した最急降下曲線でもある（1696）

$$\begin{pmatrix} x \\ y \end{pmatrix} = \begin{cases} a(t - sin\theta) \\ a(1 - cos\theta) \end{cases}$$

   $(\theta = 0 \rightarrow 2\pi)$




```
cycloid=: 3 : 0
pd 'reset'
t=. steps 0 2p1 100
a0=. t - sin t
a1=.  -. cos t
plot a0;a1
pd 'show'
)
```

2. レムニスケート曲線

   ガウスのレムニスケート曲線。1793 年のガウスの数学日記にレムニスケートが書かれている。レムニスケートとは花のリボンのこと

   遠目に見ていたガウスの難解曲線を描くことができた。

   - 直交座標

$$(x^2 + y^2)^2 - 2a^2(x^2 - y^2) = 0$$

- 極座標

$$r^2 = a^2 cos(2\theta)$$

- パラメータ表示

$$\begin{pmatrix} x \\ y \end{pmatrix} = \begin{cases} \dfrac{asin\theta cos\theta}{1+sin^2\theta} = \dfrac{t(t^2+1)}{1+t^4} \\ \dfrac{acos\theta}{1+sin^2\theta} = \dfrac{-t(t^2-1)}{1+t^4} \end{cases}$$

$(\theta = -2\pi \leftrightarrow 2\pi)$




```
lemniscate=: 3 : 0
pd 'reset'
t=. steps _2p1 2p1 100
t1=. >: ^&2 sin t
a0=. (cos t)  % t1
a1=.  ((sin t)*cos t) % t1
plot a0;a1
pd 'show'
)
```

3　媒介変数と 3D グラフィックス

3.1　1 変数の線

（a）交流

$$\begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{cases} cos\theta \\ sin\theta \\ \theta \end{cases}$$

2D に時間 t を考慮して *3D* にすると次のような図になる。

```
plot tmp; (cos;sin) tmp=.steps  0 12p1 1080
```



- 電力の交流の模式である。
- 図は x 軸を θ としている
- xyz の順は次で変えられる

```
plot 1 0 2 { tmp; (sin;cos)tmp=: steps 0 12p1 1080
plot 1 2 0 { tmp; (sin;cos)tmp=: steps 0 12p1 1080
```

（b）図




```
test_param3d_0=: 3 : 0
t=. steps _1 1 300
a0=. cos 4p1 *t
a1=. sin 2p1 *t
a2=. sin 1p1 *t
ax=: a0 ;a1;a2
plot ax
)
test_param3d_0 ''
```

3.2　3D の面

3D の面のグラフィックスになると、ネット検索では gnuplot や描画ツールにヒットすることが多くなる。ツールの入力パラメータを抜き出したものもあるが、ツールや J の癖もあり、素直には描けないことが多い。ツールによっては陰陽関数を入力すると 3D グラフを描くものもある。

（a）図 1






```
test_param3d_1 =: 3 : 0

t=. steps 0 6p1 100  NB. theta

a0=. cos  t        NB. x axis
a1=. sin  t        NB. y
a2=. 0.1 * t       NB. z
ax=: a0 ;a1;a2     NB. x;y;z
pd ax
pd 'show'
)
```

```
test_param3d_2 =: 3 : 0
t=. steps 0 2p1 50
s=. steps _1 1 50
a0=. s */ cos  t
a1=. s */ sin  t
a2=. ($ a1) $ 0.2 * t
ax=: a0 ; a1 ; a2
pd 'type wire'
pd 'color blue'
pd  ax
pd 'show'
)
```

（b）球面

$$\begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{cases} cos\theta \cdot cos\phi \\ sin\theta \cdot cos\phi \\ sin\phi \end{cases}$$




```
test_param3d_6 =: 3 : 0
t=. steps _1p1 1p1 100
s=. steps _2p1 2p1 100
a0=. (cos s) */ cos t
a1=. (sin s) */ cos t
a2=. ($ a1) $ sin t
ax=: a0 ; a1 ; a2
pd ' color blue'
pd 'type wire'
pd ax
pd 'show'
)
```

（c）トーラス

$$\begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{cases} cos\theta \cdot cos\frac{3}{2} + \phi \\ sin\theta \cdot \frac{3}{2} + cos\phi \\ sin\theta \end{cases} \quad (\theta, \phi = 0 \rightarrow 2\pi)$$




```
test_param3d_5 =: 3 : 0
NB. toras
t=. steps 0 2p1 100
s=. steps 0 2p1 100
a0=. (cos s) */ 3r2 + cos t
a1=. ( sin s) */ 3r2+ cos t
a2=.($ a0) $  sin t
ax=: a0 ; a1 ; a2
pd ' color green'
pd 'type wire'
pd ax
pd 'show'
)
```

付録 A

（a）レムニスケート 2

後ろの方の t のパラメータのみで描くと何故か図が完結しない

$$\begin{pmatrix} x \\ y \end{pmatrix} = \begin{cases} \dfrac{asin\theta cos\theta}{1 + sin^2\theta} = \dfrac{t(t^2+1)}{1+t^4} \\ \dfrac{acos\theta}{1+sin^2\theta} = \dfrac{-t(t^2-1)}{1+t^4} \end{cases}$$

$(\theta = -2\pi \leftrightarrow 2\pi)$




```
lemniscate2=: 3 : 0
pd 'reset'
t=. steps _2p1 2p1 200
t1=. >: ^&4 t
a0=.(t* (>: *: t))  % t1
a1=.  (- t*(<: *: t)) % t1
plot a0;a1
pd 'show'
)
```

（b）サイクロイドの微分

微分法は次を参照した。

http://www.geisya.or.jp/~mwm48961/kou2/d_para1.html

媒介変数は個別に微分できる

$$\frac{dy}{dx} = \frac{\frac{dy}{dt}}{\frac{dx}{dt}} = \frac{sint}{1-cost}$$

分子分母を yx として描いてみた。




```
cycloid_diff=: 3 : 0
pd 'reset'
t=. steps 0 2p1 100
a0=. t - sin t
a1=.  -. cos t
pd a0;a1
NB. diff
pd 'color red'
d1=. sin t
pd a1;d1
pd 'show'
)
```

3. アステロイドの微分

$$\begin{pmatrix} x \\ y \end{pmatrix} = \begin{cases} acos^3 t \\ asin^3 t \end{cases}$$

$(\theta = -2\pi \leftrightarrow 2\pi)$

$$\frac{dy}{dx} = \frac{\frac{dy}{dt}}{\frac{dx}{dt}} = \frac{3a \cdot cos^2 t(-sint)}{3a \cdot sin^2 t \cdot cost} = -tant$$

分子分母を yx として描いてみた。tan のグラフは暴れるので $\frac{dy}{dx}$ を分かち書きしてみた。




```
cycloid_diff=: 3 : 0
pd 'reset'
t=. steps 0 2p1 100
a0=. t - sin t
a1=.  -. cos t
pd a0;a1
NB. diff
pd 'color red'
d1=. sin t
pd a1;d1
pd 'show'
)
```